



گروه ساقه

کار با ستونهای محاسباتی  
CALCULATED COLUMNS

نشانی اینترنتی : <http://saghehgroup.com>

شماره تماس : ۰۲۱-۸۸۲۱۱۳۶۸

## کار با ستونهای محاسباتی

ستونهای محاسبه ای ستونهایی هستند که می توان به کمک آنها مقادیر ستونهای دیگر را ترکیب کرد. در ادامه روند کار با این ستونها و چندین مثال متنوع بحث شرح داده می شود.

ابتدا جهت ایجاد ستون محاسبه ای ، در تنظیمات لیست بر روی ایجاد ستون کلیک کنید. سپس نوع یا Type آن را بر روی Calculated یا محاسبه ای قرار دهید.

در کادری به نام Formula یا فرمول که برای شما باز خواهد شد، میتوانید فرمول مورد نظر خود را بنویسید. و در بخش زیرین نیز می توانید نوع داده بازگشتی مورد نظر خود را تعیین کنید.

توجه: در زمان ایجاد کردن ستون محاسبه ای حتما از نوع صحیح بازگشتی آن اطمینان حاصل کنید. حال در این بخش نمونه هایی از فرمول ها و مثال هایی برای شما جمع آوری کرده ایم:

## فرمول های تاریخ و زمان

برای اضافه کردن عدد به تاریخ از عملگر (+) استفاده کنید. برای کم کردن مقدار یک عدد از عملگر (-) استفاده کنید.

توضیحات	Formula	Column2 (Number)	Column1 (Date)
سه روز به ۲۰۱۶/۰۶/۰۹ اضافه میکند (۲۰۱۶/۰۹/۰۹)	=[Column1]+[Column2]	3	2016/09/06
۴روز از ۲۰۱۶/۱۰/۱۲ کم میکند (۲۰۱۶/۱۰/۰۸)	=[Column1]-4		2016/10/12

## اضافه یا کم کردن روز، ماه و یا سال از تاریخ

از توابع DATE , YEAR,MONTH,DAY استفاده کنید.

توضیحات	Formula	Column2 (Number)	Column1 (Date)
سه ماه به ۲۰۱۶/۰۹/۰۶ اضافه میکند (۲۰۱۶/۱۲/۰۶)	=DATE(YEAR([Column1])+[Column2],MONTH([Column1]),DAY([Column1]))	3	2016/09/06
سه سال و پنج ماه و یک روز به ۲۰۱۶/۰۲/۱۰ اضافه میکند (۲۰۱۹/۰۷/۱۱)	=DATE(YEAR([Column1])+3,MONTH([Column1])+[Column2],DAY([Column1])+1)	5	2016/02/10

## محاسبه اختلاف بین دو تاریخ

از تابع DATEDIF استفاده کنید، نوع داده بازگشتی باید یک عدد باشد.

توضیحات	Formula	Column2	Column1
تعداد روزهای بین دو تاریخ را بر میگرداند (۱۶۶)	=DATEDIF([Column1], [Column2],"d")	2016/06/15	2016/01/ 01
تعداد ماه های بین دو تاریخ را بر میگرداند (۱۷)	=DATEDIF([Column1], [Column2],"m")	2016/01/01	2014/08/ 01

## محاسبه اختلاف بین دو زمان

از تابع INT برای کم کردن یک date/time از دیگری استفاده کنید، سپس آن عدد را در عدد واحد زمان (ساعت، دقیقه، ثانیه) در روز ضرب کنید. نوع داده بازگشتی باید یک عدد باشد.

توضیحات	Formula	Column2	Column1
تعداد ساعات بین دو زمان (۲۸)	=INT(((Column2)- [Column1])*24)	2016/06/10 3:30 PM	2016/06/09 10:35 AM
تعداد دقیقه ها بین دو زمان (۱۷۳۵)	=INT(((Column2)- [Column1])*1440)	2016/06/10 3:30 PM	2016/06/09 10:35 AM
تعداد ثانیه ها بین دو زمان (۱۰۴۱۰۰)	=INT(((Column2)- [Column1])*86400)	2016/06/10 3:30 PM	2016/06/09 10:35 AM

## نمایش تاریخ بر اساس روز هفته

برای تبدیل تاریخ به متن به جهت روز هفته، از توابع TEXT و WEEKDAY استفاده کنید. نوع داده بازگشتی باید یک خط متن باشد.

توضیحات	Formula	Column1
روز هفته برای تاریخ را محاسبه کرده و نام کامل آن را بر میگرداند (Saturday)	=TEXT(WEEKDAY([Column1]), "dddd")	2016/04/16
روز هفته برای تاریخ را محاسبه کرده و نام مخفف آن را بر میگرداند (Thu)	=TEXT(WEEKDAY([Column1]), "ddd")	2016/04/21

## فرمول های ریاضی (عددی)

شما میتوانید از فرمول های جاری برای اجرای محاسبات ریاضی متنوعی استفاده کنید ، مانند جمع، تفریق، ضرب، تقسیم اعداد؛ محاسبه میانگین اعداد و انجام مقایسه های بزرگتر و کوچکتر.

## جمع یا تفریق اعداد

برای اضافه کردن اعداد در دو ستون یا بیشتر در ردیف از عملگر جمع (+) یا تفریق (-) استفاده کنید.

توضیحات	Formula	ColumnC	ColumnB	ColumnA
جمع ۱۵+۱۰+۵ با هم (۳۰)	=[ColumnA]+[ColumnB]+[ColumnC]	5	10	15
تفریق ۱۵-۱۰-۵ (۰)	=[ColumnA]-[ColumnB]-[ColumnC]	5	10	15

## ضرب یا تقسیم اعداد

توضیحات	Formula	Column2	Column1
ضرب ۵ در ۲ (۱۰۰)	=[Column1]*[Column2]	2	5
تقسیم ۱۵۰۰۰ بر ۱۲ (۱۲۵۰)	=[Column1]/ [Column2]	12	15000

## محاسبه میانگین اعداد

برای محاسبه کردن میانگین اعداد در دو ستون یا بیشتر در ردیف از تابع AVERAGE استفاده کنید.

توضیحات	Formula	Column 3	Column 2	Column 1
میانگین اعداد در سه ستون اول (۵)	=AVERAGE([Column1],[Column2],[Column3])	4	5	6
اگر ستون ۱ از ستون ۲ بزرگتر بود، میانگین تفریق ستونها و ستون ۳ را محاسبه کند. در غیر این صورت میانگین عدد ۱۰ و ستون ۳ را حساب کند. (۲,۵)	=AVERAGE(IF([Column1]>[Column2],[Column1]-[Column2],10), [Column3])	4	5	6

## محاسبه کوچکترین یا بزرگترین عدد در یک بازه

برای محاسبه کوچکترین یا بزرگترین عدد در دو ستون یا بیشتر از توابع MIN و MAX استفاده کنید.

توضیحات	Formula	Column3	Column2	Column1
کوچکترین عدد (۷)	=MIN([Column1],[Column2],[Column3])	9	7	10
بزرگترین عدد (۱۰)	=MAX([Column1],[Column2],[Column3])	9	7	10

## محاسبه اختلاف بین دو عدد بر حسب درصد

از عملگرهای تفریق (-) و تقسیم (/) و تابع ABS استفاده کنید.

توضیحات	Formula	Column2	Column1
تغییر بر حسب درصد (۶,۷۵٪)	=([Column2]-[Column1])/ABS([Column1])	2500	2342

افزایش یا کاهش دادن یک عدد بر حسب درصد

برای این محاسبه از عملگر (%) استفاده کنید.

توضیحات	Formula	Column2	Column1
افزایش عدد در ستون یک با ۵ درصد (۱۰۵)	=[Column1]*(1.05)		100
افزایش عدد در ستون یک بوسیله مقدار درصد در ستون دو: ۷٪ (۱۰۷)	=[Column1]*(1+[Column2])	۷٪ (ستون ۲ باید فیلد عددی باشد تا به درصد نمایش داده شود)	100
کاهش عدد در ستون یک بوسیله مقدار درصد در ستون دو: ۷٪ (۹۳)	=[Column1]*(1-[Column2])	۷٪ (ستون ۲ باید فیلد عددی باشد تا به درصد نمایش داده شود)	100

تعیین بزرگتر یا کوچکتر بودن عدد از عددی دیگر

جهت مقایسه فیلدهای عددی از تابع IF استفاده کنید. نوع داده بازگشتی باید یک خط متن باشد.

توضیحات	Formula	Column2	Column1
آیا ستون یک از ستون دو بزرگتر است؟ (Yes)	=IF([Column1]>[Column2], "Yes", "No")	9	15
آیا ستون یک کوچکتر یا مساوی با ستون دو است؟ (Not OK)	=IF([Column1]<=[Column2], "OK", "Not OK")	9	15

## شرط های عددی با یک تابع AND یا OR

با استفاده از توابع AND یا OR میتوانید فرمول های شرطی پیچیده تری ایجاد کنید.

توضیحات	Formula	Column3	Column2	Column1
اگر ۱۵ بزرگتر از ۹ است و ۱۵ کوچکتر از ۸ است "OK" را برگردان. (Not OK)	=IF(AND([Column1]>[Column2], [Column1]<[Column3]), "OK", "Not OK")	8	9	15
اگر ۱۵ بزرگتر از ۹ است یا ۱۵ کوچکتر از ۸ است "Yes" را برگردان. (Yes)	=IF(OR([Column1]>[Column2], [Column1]<[Column3]), "Yes", "No")	8	9	15

## فرمول های متنی

شما میتوانید از فرمول های جاری جهت ترکیب متن ها استفاده کنید، مانند ترکیب یا الحاق مقادیر چندین ستون، مقایسه محتوی ستونها، و حذف کاراکترها یا فاصله ها.

## تغییر قالب متن

برای تغییر قالب متن از توابع UPPER, LOWER یا PROPER استفاده کنید.

توضیحات	Formula	Column1
متن را به حروف بزرگ تغییر می دهد (JOHN DOE)	=UPPER([Column1])	John Doe
متن را به حروف کوچک تغییر می دهد (john doe)	=LOWER([Column1])	John Doe
حروف اول متن را به بزرگ تغییر می دهد (Jane Doe)	=PROPER([Column1])	jane doe

## ترکیب مقادیر ستون متنی با یکدیگر

برای ترکیب نام و نام خانوادگی، از عملگر (&) یا تابع CONCATENATE استفاده کنید.

توضیحات	Formula	Column2	Column1
دو متن را با هم ترکیب می کند (CarlosCarvallo)	=[Column1]&[Column2]	Carvallo	Carlos
دو متن را با هم ترکیب می کند، با یک فاصله آنها را جدا می کند (Carlos Carvallo)	=[Column1]&" &[Column2]	Carvallo	Carlos
دو متن را با هم ترکیب می کند، با یک فاصله و یک ویرگول آنها را جدا می کند (Carvallo, Carlos)	=[Column2]&" &[Column1]	Carvallo	Carlos

## ترکیب متن و اعداد از ستونهای مختلف

برای ترکیب متن و اعداد، از تابع CONCATENATE، عملگر (&)، یا تابع TEXT و عملگر (&) استفاده کنید.

توضیحات	Formula	Column2	Column1
محتوا را در یک متن ترکیب می کند (Yang sold 28 units.)	= [Column1] & " sold & [Column2] & " units."	28	Yang
محتوا را در یک متن ترکیب می کند. توجه کنید که تابع TEXT فرمت متن موجود در ستون ۲ را به جای مقدار ۰,۴ اصلاح میکند (Dubois sold 40% of the total sales.)	= [Column1] & " sold & TEXT([Column2], "0%") & of the total sales."	40%	Dubois

## ترکیب متن با یک تاریخ یا زمان

برای ترکیب متن با یک تاریخ یا زمان، از تابع TEXT و عملگر (&) استفاده کنید.

توضیحات	Formula	Column2	Column1
متن را با یک تاریخ ترکیب می کند (Payment due: June 20, 2016)	= "Payment due: & TEXT([Column2], " mmmm dd, yyyy")	2016/06/20	
متن و تاریخ را از ستونهای متفاوت درون یک ستون ترکیب میکند (Billing Date Jun-05-2016)	= [Column1] & & TEXT([Column2], " mmm-dd-yyyy")	2016/06/05	Billing Date

## تعیین اینکه مقدار یک ستون با بخشی از آن با متن مشخصی همخوانی دارد

برای تعیین اینکه مقدار یک ستون یا بخشی از آن با متن مشخصی همخوانی دارد یا خیر، از توابع IF، SEARCH، FIND و ISNUMBER استفاده کنید.

توضیحات	Formula	Column1
تعیین می کند که آیا ستون یک Urgent است یا خیر (Yes)	=IF([Column1]="Urgent", "Yes", "No")	Urgent
تعیین می کند که آیا ستون یک شامل حرف s است یا خیر (OK)	=IF(ISNUMBER(FIND("s",[Column1])), "OK", "Not OK")	Standard
تعیین می کند که آیا ستون یک شامل BD است یا خیر (Yes)	=ISNUMBER(FIND("BD",[Column1]))	BD123

## حذف کاراکترها از متن

برای حذف کاراکترها از متن، از توابع LEFT، RIGHT و LEN استفاده کنید.

توضیحات	Formula	Column1
۷ کاراکتر بر میگرداند (2-9)، شروع از سمت چپ (Vitamin)	=LEFT([Column1],LEN([Column1])-2)	Vitamin A
۲ کاراکتر بر میگرداند (8-10)، شروع از سمت راست (B1)	=RIGHT([Column1], LEN([Column1])-8)	Vitamin B1

## حذف فاصله ها از ابتدا و انتهای یک ستون

برای حذف فاصله ها از یک ستون، از تابع TRIM استفاده کنید.

توضیحات	Formula	Column1
فاصله ها را از ابتدا و انتهای متن حذف میکند (Hello there!)	=TRIM([Column1])	Hello there!

در ادامه برخی از توابع مورد استفاده در این نوع ستونها را برای شما آماده کرده ایم



مثال	توضیح	متن
LEN("Hello") {5}	The length of Text	LEN (Text)
LEFT("I quite like camping", 6) {"I quit"}	Return X characters from the left	LEFT (Text, Number)
RIGHT("I quite like camping",4) {"ping"}	Returns X characters from the right	RIGHT (Text, Number)
MID("I quite like camping", 9, 11) {"like campin"}	Returns Number2 characters from the middle of Text, starting at Number1	MID (Text, Num1, Num2)
SEARCH("Banana", "Banana Banana", 4) {8}	Returns the index of Text1 within Text2, starting the search at index Number	SEARCH (Text1, Text2, Num)
LOWER("uRGEnT") {"urgent"}	Text in lower case	LOWER (Text)
UPPER("uRGEnT") {"URGENT"}	Text in upper case	UPPER (Text)
PROPER("uRGEnT TICKet") {"Urgent Ticket"}	Capitalize first letter of each word	PROPER (Text)
TRIM(" some text ") {"some text"}	Removes spaces from the start and end	TRIM (Text)
CLEAN("Line1?? Line2??") {"Line1 Line2"}	Returns Text without non-printable characters added by clipboard or similar	CLEAN (Text)
REPLACE("Hello",2,4,"i") {"Hi"}	Replaces Number2 characters starting at Number1 from Text1 with Text2	REPLACE (T1, N1, N2, T2)
CONCATENATE("A","B","C") {"ABC"}	Puts the values together into one string	CONCATENATE (T1, T2, ...)
DOLLAR(12.567,2) {"\$12.57"}	Converts a number to a currency text, with the given number of decimals	DOLLAR (Number, Decimals)
EXACT("Hi","hi") {False}	Are the two text values identical?	EXACT (Text1, Text2)

FIXED(1057.26,1,TRUE) {"1057.3"},FIXED(1057.26,0,FALSE) {"1,057"}	Returns Number with the given number of decimals as text (commas optional)	FIXED (Num, Dec, NoCommas)
REPT("Hi",3) {"HiHiHi"}	Repeats Text the given Number of times	REPT (Text, Number)
TEXT([Created], "yyyy mm") {"2011 03"}	Converts Value to a Text value, using Format (see 'TEXT Formats' overleaf)	TEXT (Value, Format)
		محاسباتی
ABS(-1) {1}, ABS(1) {1}	Makes a number positive if it is negative	ABS (Number)
EVEN(0.5) {2}, EVEN(-1) {-2}	Rounds Number up to the nearest even number	EVEN (Number)
FACT(5) {5*4*3*2*1 = 120}	Returns the factorial of a number ("x!" on a calculator)	FACT (Number)
MOD(5, 4) {1}, MOD(2010, 20) {10}	Returns the remainder of Number1 divided by Number2	MOD (Number1, Number2)
ODD(1.5) {3}, ODD(-2) {-3}	Rounds Number up to the nearest odd number	ODD (Number)
PI() {3.14159265358979}	Returns Pi to 15 decimal places	PI ()
ROUND(327.598, 1) {327.6}	Rounds Number1 to Number2 decimals	ROUND (Number1, Number2)
ROUNDDOWN(327.598, 1) {327.5}	Rounds Number1 to Number2 decimals, always rounding down.	ROUNDDOWN (Num1, Num2)
ROUNDUP(52.0001, 0) {53}	Rounds Number1 to Number2 decimals, always rounding up.	ROUNDUP (Num1, Num2)
SIGN(-5.9786) {-1}, SIGN(20678) {1}, SIGN(0) {0}	Returns -1 for negative numbers, 1 for positive, and 0 when 0	SIGN (Number)
SQRT(64) {8}	Returns the square root	SQRT (Number)
TRUNC(37.999999) {37}	Returns Number with decimals removed	TRUNC (Number)
VALUE("57") {57}, VALUE("00:05") {00:05}	Converts Text to a Number, Date or Time, according to its format	VALUE (Text)

AVERAGE(10, 0, "", "0") {5}, AVERAGE(5, 0, TRUE) {2.5}	Averages the Numbers, ignoring nonNumber values	AVERAGE (Num1, Num2, ...)
AVERAGEA(10, 0, "", "0") {2.5}, AVERAGEA(5, 0, TRUE) {2}	Averages the Numbers, non-Number values are interpreted	AVERAGEA (Num1, Num2, ...)
COUNT(10, 0, "", "0") {2}, COUNT(5, 0, TRUE) {2}	Averages the Numbers, ignoring nonNumber values	COUNT (Num1, Num2, ...)
COUNTA(10, 0, "", "0") {4}, COUNTA(5, 0, TRUE) {3}	Counts all values, including empty text (""), ignoring empty columns	COUNTA (Value1, Value2, ...)
MAX(0, 2, "76", 100) {100}	Gets the largest of the numbers, including Text fields containing numbers	MAX (Number1, Number2, ...)
MAXA(0, 2, "76", 100, "Telephone", TRUE) {100}	Gets the largest of the numbers, including non-number values	MAXA (Number1, Number2, ...)
MIN(0, 2, "76", 100) {0}	Gets the smallest of the numbers, including Text fields containing numbers	MIN (Number1, Number2, ...)
MINA(0, 2, "76", 100, "Telephone", TRUE) {0}	Gets the smallest of the numbers, including non-number values	MINA (Number1, Number2, ...)
SUM(0, 2, "76", 100, TRUE) {179}	Returns the total of all Numbers and number-like values	SUM (Number1, Number2, ...)
		تاریخ و زمان
08:35:21	TEXT([DateTimeField],"hh:mm:ss")	Time only
Tuesday Tue	TEXT([DateField],"dddd") TEXT([DateField],"ddd")	Weekday
September Sep	TEXT([DateField],"mmmm") TEXT([DateField],"mmm")	Month
2011 11	TEXT([DateField],"yyyy") TEXT([DateField],"yy")	Year
September 17, 2011	TEXT([DateField],"mmmm dd, yyyy")	Combinations
AM	Shows whether a time is AM or PM IF(HOUR([Time]) < 12,"AM","PM")	AM/PM

Day	Shows whether a time is day or night IF(AND(HOUR([Time])>6,HOUR([Time])<18),"Day","Night")	Day/Night
5	Shows the week number (US Style) ROUNDDOWN(([Date]- DATE(YEAR([Date]),1,1)+WEEKDAY(DATE(YEAR([Date]),1,1)- WEEKDAY([Date])+1)/7,0)+1	Week number
Spring	Shows which season a date falls in. Takes into account one month offset from quarter. CHOOSE(INT((MOD(MONTH(When)+1,12)/4))+1,"Winter","Spring","Summer","Autumn")	Season
YEAR([Created]) {2011}	Returns the year number from a date	YEAR (Date)
MONTH([Created]) {3}	Returns the month number from a date	MONTH (Date)
WEEKDAY([Created]) {2}	Returns the weekday number from a date or datetime (0 is Sunday)	WEEKDAY (Date)
DAY([Created]) {29}	Returns the day number from a date	DAY (Date)
HOUR([Created]) {13}	Returns the hour (uses 24h clock)	HOUR (DateTime)
MINUTE([Created]) {51}	Returns the minute from a datetime	MINUTE (DateTime)
SECOND([Created]) {26}	Returns the seconds from a datetime	SECOND (DateTime)
DATE(2011, 3, 29) {3/29/2011}	Creates a date from the 3 numbers	DATE (Year, Month, Day)
DATEDIF("3/29/2011","2/1/2011","M") {1}	The difference between two dates (Unit: "Y" Years, "M" Months, "D" Days, and more)	DATEDIF (Date1, Date2, Unit)
DATEVALUE("3/29/2011") {3/29/2011}	Converts Text to a Date	DATEVALUE (Text)
TIME(19,45,21) {1/1/1900 7:45:21 PM}	Creates a time from the 3 numbers	TIME (Hour, Minute, Second)
TIMEVALUE("19:45") {1/1/1900 7:45 PM}	Converts Text to a time value	TIMEVALUE (Text)